creation_sourcecode: // SPDX-License-Identifier: AGPL-3.0
  pragma solidity =0.8.18 >=0.8.18 ^0.8.0;
  
  // lib/openzeppelin-contracts/contracts/token/ERC20/IERC20.sol
  
  // OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)
  
  /**
   * @dev Interface of the ERC20 standard as defined in the EIP.
   */
  interface IERC20 {
      /**
       * @dev Emitted when `value` tokens are moved from one account (`from`) to
       * another (`to`).
       *
       * Note that `value` may be zero.
       */
      event Transfer(address indexed from, address indexed to, uint256 value);
  
      /**
       * @dev Emitted when the allowance of a `spender` for an `owner` is set by
       * a call to {approve}. `value` is the new allowance.
       */
      event Approval(address indexed owner, address indexed spender, uint256 value);
  
      /**
       * @dev Returns the amount of tokens in existence.
       */
      function totalSupply() external view returns (uint256);
  
      /**
       * @dev Returns the amount of tokens owned by `account`.
       */
      function balanceOf(address account) external view returns (uint256);
  
      /**
       * @dev Moves `amount` tokens from the caller's account to `to`.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * Emits a {Transfer} event.
       */
      function transfer(address to, uint256 amount) external returns (bool);
  
      /**
       * @dev Returns the remaining number of tokens that `spender` will be
       * allowed to spend on behalf of `owner` through {transferFrom}. This is
       * zero by default.
       *
       * This value changes when {approve} or {transferFrom} are called.
       */
      function allowance(address owner, address spender) external view returns (uint256);
  
      /**
       * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * IMPORTANT: Beware that changing an allowance with this method brings the risk
       * that someone may use both the old and the new allowance by unfortunate
       * transaction ordering. One possible solution to mitigate this race
       * condition is to first reduce the spender's allowance to 0 and set the
       * desired value afterwards:
       * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
       *
       * Emits an {Approval} event.
       */
      function approve(address spender, uint256 amount) external returns (bool);
  
      /**
       * @dev Moves `amount` tokens from `from` to `to` using the
       * allowance mechanism. `amount` is then deducted from the caller's
       * allowance.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * Emits a {Transfer} event.
       */
      function transferFrom(address from, address to, uint256 amount) external returns (bool);
  }
  
  // lib/openzeppelin-contracts/contracts/token/ERC20/extensions/IERC20Permit.sol
  
  // OpenZeppelin Contracts (last updated v4.9.4) (token/ERC20/extensions/IERC20Permit.sol)
  
  /**
   * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
   * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
   *
   * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
   * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
   * need to send a transaction, and thus is not required to hold Ether at all.
   *
   * ==== Security Considerations
   *
   * There are two important considerations concerning the use of `permit`. The first is that a valid permit signature
   * expresses an allowance, and it should not be assumed to convey additional meaning. In particular, it should not be
   * considered as an intention to spend the allowance in any specific way. The second is that because permits have
   * built-in replay protection and can be submitted by anyone, they can be frontrun. A protocol that uses permits should
   * take this into consideration and allow a `permit` call to fail. Combining these two aspects, a pattern that may be
   * generally recommended is:
   *
   * ```solidity
   * function doThingWithPermit(..., uint256 value, uint256 deadline, uint8 v, bytes32 r, bytes32 s) public {
   *     try token.permit(msg.sender, address(this), value, deadline, v, r, s) {} catch {}
   *     doThing(..., value);
   * }
   *
   * function doThing(..., uint256 value) public {
   *     token.safeTransferFrom(msg.sender, address(this), value);
   *     ...
   * }
   * ```
   *
   * Observe that: 1) `msg.sender` is used as the owner, leaving no ambiguity as to the signer intent, and 2) the use of
   * `try/catch` allows the permit to fail and makes the code tolerant to frontrunning. (See also
   * {SafeERC20-safeTransferFrom}).
   *
   * Additionally, note that smart contract wallets (such as Argent or Safe) are not able to produce permit signatures, so
   * contracts should have entry points that don't rely on permit.
   */
  interface IERC20Permit {
      /**
       * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
       * given ``owner``'s signed approval.
       *
       * IMPORTANT: The same issues {IERC20-approve} has related to transaction
       * ordering also apply here.
       *
       * Emits an {Approval} event.
       *
       * Requirements:
       *
       * - `spender` cannot be the zero address.
       * - `deadline` must be a timestamp in the future.
       * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
       * over the EIP712-formatted function arguments.
       * - the signature must use ``owner``'s current nonce (see {nonces}).
       *
       * For more information on the signature format, see the
       * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
       * section].
       *
       * CAUTION: See Security Considerations above.
       */
      function permit(
          address owner,
          address spender,
          uint256 value,
          uint256 deadline,
          uint8 v,
          bytes32 r,
          bytes32 s
      ) external;
  
      /**
       * @dev Returns the current nonce for `owner`. This value must be
       * included whenever a signature is generated for {permit}.
       *
       * Every successful call to {permit} increases ``owner``'s nonce by one. This
       * prevents a signature from being used multiple times.
       */
      function nonces(address owner) external view returns (uint256);
  
      /**
       * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
       */
      // solhint-disable-next-line func-name-mixedcase
      function DOMAIN_SEPARATOR() external view returns (bytes32);
  }
  
  // lib/openzeppelin-contracts/contracts/utils/Context.sol
  
  // OpenZeppelin Contracts (last updated v4.9.4) (utils/Context.sol)
  
  /**
   * @dev Provides information about the current execution context, including the
   * sender of the transaction and its data. While these are generally available
   * via msg.sender and msg.data, they should not be accessed in such a direct
   * manner, since when dealing with meta-transactions the account sending and
   * paying for execution may not be the actual sender (as far as an application
   * is concerned).
   *
   * This contract is only required for intermediate, library-like contracts.
   */
  abstract contract Context {
      function _msgSender() internal view virtual returns (address) {
          return msg.sender;
      }
  
      function _msgData() internal view virtual returns (bytes calldata) {
          return msg.data;
      }
  
      function _contextSuffixLength() internal view virtual returns (uint256) {
          return 0;
      }
  }
  
  // lib/tokenized-strategy/src/interfaces/IBaseStrategy.sol
  
  interface IBaseStrategy {
      function tokenizedStrategyAddress() external view returns (address);
  
      /*//////////////////////////////////////////////////////////////
                              IMMUTABLE FUNCTIONS
      //////////////////////////////////////////////////////////////*/
  
      function availableDepositLimit(
          address _owner
      ) external view returns (uint256);
  
      function availableWithdrawLimit(
          address _owner
      ) external view returns (uint256);
  
      function deployFunds(uint256 _assets) external;
  
      function freeFunds(uint256 _amount) external;
  
      function harvestAndReport() external returns (uint256);
  
      function tendThis(uint256 _totalIdle) external;
  
      function shutdownWithdraw(uint256 _amount) external;
  
      function tendTrigger() external view returns (bool, bytes memory);
  }
  
  // lib/tokenized-strategy-periphery/src/utils/Governance.sol
  
  contract Governance {
      /// @notice Emitted when the governance address is updated.
      event GovernanceTransferred(
          address indexed previousGovernance,
          address indexed newGovernance
      );
  
      modifier onlyGovernance() {
          _checkGovernance();
          _;
      }
  
      /// @notice Checks if the msg sender is the governance.
      function _checkGovernance() internal view virtual {
          require(governance == msg.sender, "!governance");
      }
  
      /// @notice Address that can set the default base fee and provider
      address public governance;
  
      constructor(address _governance) {
          governance = _governance;
  
          emit GovernanceTransferred(address(0), _governance);
      }
  
      /**
       * @notice Sets a new address as the governance of the contract.
       * @dev Throws if the caller is not current governance.
       * @param _newGovernance The new governance address.
       */
      function transferGovernance(
          address _newGovernance
      ) external virtual onlyGovernance {
          require(_newGovernance != address(0), "ZERO ADDRESS");
          address oldGovernance = governance;
          governance = _newGovernance;
  
          emit GovernanceTransferred(oldGovernance, _newGovernance);
      }
  }
  
  // lib/openzeppelin-contracts/contracts/token/ERC20/extensions/IERC20Metadata.sol
  
  // OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)
  
  /**
   * @dev Interface for the optional metadata functions from the ERC20 standard.
   *
   * _Available since v4.1._
   */
  interface IERC20Metadata is IERC20 {
      /**
       * @dev Returns the name of the token.
       */
      function name() external view returns (string memory);
  
      /**
       * @dev Returns the symbol of the token.
       */
      function symbol() external view returns (string memory);
  
      /**
       * @dev Returns the decimals places of the token.
       */
      function decimals() external view returns (uint8);
  }
  
  // lib/tokenized-strategy-periphery/src/AprOracle/AprOracleBase.sol
  
  abstract contract AprOracleBase is Governance {
      string public name;
  
      constructor(
          string memory _name,
          address _governance
      ) Governance(_governance) {
          name = _name;
      }
  
      /**
       * @notice Will return the expected Apr of a strategy post a debt change.
       * @dev _delta is a signed integer so that it can also represent a debt
       * decrease.
       *
       * _delta will be == 0 to get the current apr.
       *
       * This will potentially be called during non-view functions so gas
       * efficiency should be taken into account.
       *
       * @param _strategy The strategy to get the apr for.
       * @param _delta The difference in debt.
       * @return . The expected apr for the strategy.
       */
      function aprAfterDebtChange(
          address _strategy,
          int256 _delta
      ) external view virtual returns (uint256);
  }
  
  // lib/openzeppelin-contracts/contracts/interfaces/IERC4626.sol
  
  // OpenZeppelin Contracts (last updated v4.9.0) (interfaces/IERC4626.sol)
  
  /**
   * @dev Interface of the ERC4626 "Tokenized Vault Standard", as defined in
   * https://eips.ethereum.org/EIPS/eip-4626[ERC-4626].
   *
   * _Available since v4.7._
   */
  interface IERC4626 is IERC20, IERC20Metadata {
      event Deposit(address indexed sender, address indexed owner, uint256 assets, uint256 shares);
  
      event Withdraw(
          address indexed sender,
          address indexed receiver,
          address indexed owner,
          uint256 assets,
          uint256 shares
      );
  
      /**
       * @dev Returns the address of the underlying token used for the Vault for accounting, depositing, and withdrawing.
       *
       * - MUST be an ERC-20 token contract.
       * - MUST NOT revert.
       */
      function asset() external view returns (address assetTokenAddress);
  
      /**
       * @dev Returns the total amount of the underlying asset that is “managed” by Vault.
       *
       * - SHOULD include any compounding that occurs from yield.
       * - MUST be inclusive of any fees that are charged against assets in the Vault.
       * - MUST NOT revert.
       */
      function totalAssets() external view returns (uint256 totalManagedAssets);
  
      /**
       * @dev Returns the amount of shares that the Vault would exchange for the amount of assets provided, in an ideal
       * scenario where all the conditions are met.
       *
       * - MUST NOT be inclusive of any fees that are charged against assets in the Vault.
       * - MUST NOT show any variations depending on the caller.
       * - MUST NOT reflect slippage or other on-chain conditions, when performing the actual exchange.
       * - MUST NOT revert.
       *
       * NOTE: This calculation MAY NOT reflect the “per-user” price-per-share, and instead should reflect the
       * “average-user’s” price-per-share, meaning what the average user should expect to see when exchanging to and
       * from.
       */
      function convertToShares(uint256 assets) external view returns (uint256 shares);
  
      /**
       * @dev Returns the amount of assets that the Vault would exchange for the amount of shares provided, in an ideal
       * scenario where all the conditions are met.
       *
       * - MUST NOT be inclusive of any fees that are charged against assets in the Vault.
       * - MUST NOT show any variations depending on the caller.
       * - MUST NOT reflect slippage or other on-chain conditions, when performing the actual exchange.
       * - MUST NOT revert.
       *
       * NOTE: This calculation MAY NOT reflect the “per-user” price-per-share, and instead should reflect the
       * “average-user’s” price-per-share, meaning what the average user should expect to see when exchanging to and
       * from.
       */
      function convertToAssets(uint256 shares) external view returns (uint256 assets);
  
      /**
       * @dev Returns the maximum amount of the underlying asset that can be deposited into the Vault for the receiver,
       * through a deposit call.
       *
       * - MUST return a limited value if receiver is subject to some deposit limit.
       * - MUST return 2 ** 256 - 1 if there is no limit on the maximum amount of assets that may be deposited.
       * - MUST NOT revert.
       */
      function maxDeposit(address receiver) external view returns (uint256 maxAssets);
  
      /**
       * @dev Allows an on-chain or off-chain user to simulate the effects of their deposit at the current block, given
       * current on-chain conditions.
       *
       * - MUST return as close to and no more than the exact amount of Vault shares that would be minted in a deposit
       *   call in the same transaction. I.e. deposit should return the same or more shares as previewDeposit if called
       *   in the same transaction.
       * - MUST NOT account for deposit limits like those returned from maxDeposit and should always act as though the
       *   deposit would be accepted, regardless if the user has enough tokens approved, etc.
       * - MUST be inclusive of deposit fees. Integrators should be aware of the existence of deposit fees.
       * - MUST NOT revert.
       *
       * NOTE: any unfavorable discrepancy between convertToShares and previewDeposit SHOULD be considered slippage in
       * share price or some other type of condition, meaning the depositor will lose assets by depositing.
       */
      function previewDeposit(uint256 assets) external view returns (uint256 shares);
  
      /**
       * @dev Mints shares Vault shares to receiver by depositing exactly amount of underlying tokens.
       *
       * - MUST emit the Deposit event.
       * - MAY support an additional flow in which the underlying tokens are owned by the Vault contract before the
       *   deposit execution, and are accounted for during deposit.
       * - MUST revert if all of assets cannot be deposited (due to deposit limit being reached, slippage, the user not
       *   approving enough underlying tokens to the Vault contract, etc).
       *
       * NOTE: most implementations will require pre-approval of the Vault with the Vault’s underlying asset token.
       */
      function deposit(uint256 assets, address receiver) external returns (uint256 shares);
  
      /**
       * @dev Returns the maximum amount of the Vault shares that can be minted for the receiver, through a mint call.
       * - MUST return a limited value if receiver is subject to some mint limit.
       * - MUST return 2 ** 256 - 1 if there is no limit on the maximum amount of shares that may be minted.
       * - MUST NOT revert.
       */
      function maxMint(address receiver) external view returns (uint256 maxShares);
  
      /**
       * @dev Allows an on-chain or off-chain user to simulate the effects of their mint at the current block, given
       * current on-chain conditions.
       *
       * - MUST return as close to and no fewer than the exact amount of assets that would be deposited in a mint call
       *   in the same transaction. I.e. mint should return the same or fewer assets as previewMint if called in the
       *   same transaction.
       * - MUST NOT account for mint limits like those returned from maxMint and should always act as though the mint
       *   would be accepted, regardless if the user has enough tokens approved, etc.
       * - MUST be inclusive of deposit fees. Integrators should be aware of the existence of deposit fees.
       * - MUST NOT revert.
       *
       * NOTE: any unfavorable discrepancy between convertToAssets and previewMint SHOULD be considered slippage in
       * share price or some other type of condition, meaning the depositor will lose assets by minting.
       */
      function previewMint(uint256 shares) external view returns (uint256 assets);
  
      /**
       * @dev Mints exactly shares Vault shares to receiver by depositing amount of underlying tokens.
       *
       * - MUST emit the Deposit event.
       * - MAY support an additional flow in which the underlying tokens are owned by the Vault contract before the mint
       *   execution, and are accounted for during mint.
       * - MUST revert if all of shares cannot be minted (due to deposit limit being reached, slippage, the user not
       *   approving enough underlying tokens to the Vault contract, etc).
       *
       * NOTE: most implementations will require pre-approval of the Vault with the Vault’s underlying asset token.
       */
      function mint(uint256 shares, address receiver) external returns (uint256 assets);
  
      /**
       * @dev Returns the maximum amount of the underlying asset that can be withdrawn from the owner balance in the
       * Vault, through a withdraw call.
       *
       * - MUST return a limited value if owner is subject to some withdrawal limit or timelock.
       * - MUST NOT revert.
       */
      function maxWithdraw(address owner) external view returns (uint256 maxAssets);
  
      /**
       * @dev Allows an on-chain or off-chain user to simulate the effects of their withdrawal at the current block,
       * given current on-chain conditions.
       *
       * - MUST return as close to and no fewer than the exact amount of Vault shares that would be burned in a withdraw
       *   call in the same transaction. I.e. withdraw should return the same or fewer shares as previewWithdraw if
       *   called
       *   in the same transaction.
       * - MUST NOT account for withdrawal limits like those returned from maxWithdraw and should always act as though
       *   the withdrawal would be accepted, regardless if the user has enough shares, etc.
       * - MUST be inclusive of withdrawal fees. Integrators should be aware of the existence of withdrawal fees.
       * - MUST NOT revert.
       *
       * NOTE: any unfavorable discrepancy between convertToShares and previewWithdraw SHOULD be considered slippage in
       * share price or some other type of condition, meaning the depositor will lose assets by depositing.
       */
      function previewWithdraw(uint256 assets) external view returns (uint256 shares);
  
      /**
       * @dev Burns shares from owner and sends exactly assets of underlying tokens to receiver.
       *
       * - MUST emit the Withdraw event.
       * - MAY support an additional flow in which the underlying tokens are owned by the Vault contract before the
       *   withdraw execution, and are accounted for during withdraw.
       * - MUST revert if all of assets cannot be withdrawn (due to withdrawal limit being reached, slippage, the owner
       *   not having enough shares, etc).
       *
       * Note that some implementations will require pre-requesting to the Vault before a withdrawal may be performed.
       * Those methods should be performed separately.
       */
      function withdraw(uint256 assets, address receiver, address owner) external returns (uint256 shares);
  
      /**
       * @dev Returns the maximum amount of Vault shares that can be redeemed from the owner balance in the Vault,
       * through a redeem call.
       *
       * - MUST return a limited value if owner is subject to some withdrawal limit or timelock.
       * - MUST return balanceOf(owner) if owner is not subject to any withdrawal limit or timelock.
       * - MUST NOT revert.
       */
      function maxRedeem(address owner) external view returns (uint256 maxShares);
  
      /**
       * @dev Allows an on-chain or off-chain user to simulate the effects of their redeemption at the current block,
       * given current on-chain conditions.
       *
       * - MUST return as close to and no more than the exact amount of assets that would be withdrawn in a redeem call
       *   in the same transaction. I.e. redeem should return the same or more assets as previewRedeem if called in the
       *   same transaction.
       * - MUST NOT account for redemption limits like those returned from maxRedeem and should always act as though the
       *   redemption would be accepted, regardless if the user has enough shares, etc.
       * - MUST be inclusive of withdrawal fees. Integrators should be aware of the existence of withdrawal fees.
       * - MUST NOT revert.
       *
       * NOTE: any unfavorable discrepancy between convertToAssets and previewRedeem SHOULD be considered slippage in
       * share price or some other type of condition, meaning the depositor will lose assets by redeeming.
       */
      function previewRedeem(uint256 shares) external view returns (uint256 assets);
  
      /**
       * @dev Burns exactly shares from owner and sends assets of underlying tokens to receiver.
       *
       * - MUST emit the Withdraw event.
       * - MAY support an additional flow in which the underlying tokens are owned by the Vault contract before the
       *   redeem execution, and are accounted for during redeem.
       * - MUST revert if all of shares cannot be redeemed (due to withdrawal limit being reached, slippage, the owner
       *   not having enough shares, etc).
       *
       * NOTE: some implementations will require pre-requesting to the Vault before a withdrawal may be performed.
       * Those methods should be performed separately.
       */
      function redeem(uint256 shares, address receiver, address owner) external returns (uint256 assets);
  }
  
  // lib/openzeppelin-contracts/contracts/token/ERC20/ERC20.sol
  
  // OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/ERC20.sol)
  
  /**
   * @dev Implementation of the {IERC20} interface.
   *
   * This implementation is agnostic to the way tokens are created. This means
   * that a supply mechanism has to be added in a derived contract using {_mint}.
   * For a generic mechanism see {ERC20PresetMinterPauser}.
   *
   * TIP: For a detailed writeup see our guide
   * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
   * to implement supply mechanisms].
   *
   * The default value of {decimals} is 18. To change this, you should override
   * this function so it returns a different value.
   *
   * We have followed general OpenZeppelin Contracts guidelines: functions revert
   * instead returning `false` on failure. This behavior is nonetheless
   * conventional and does not conflict with the expectations of ERC20
   * applications.
   *
   * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
   * This allows applications to reconstruct the allowance for all accounts just
   * by listening to said events. Other implementations of the EIP may not emit
   * these events, as it isn't required by the specification.
   *
   * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
   * functions have been added to mitigate the well-known issues around setting
   * allowances. See {IERC20-approve}.
   */
  contract ERC20 is Context, IERC20, IERC20Metadata {
      mapping(address => uint256) private _balances;
  
      mapping(address => mapping(address => uint256)) private _allowances;
  
      uint256 private _totalSupply;
  
      string private _name;
      string private _symbol;
  
      /**
       * @dev Sets the values for {name} and {symbol}.
       *
       * All two of these values are immutable: they can only be set once during
       * construction.
       */
      constructor(string memory name_, string memory symbol_) {
          _name = name_;
          _symbol = symbol_;
      }
  
      /**
       * @dev Returns the name of the token.
       */
      function name() public view virtual override returns (string memory) {
          return _name;
      }
  
      /**
       * @dev Returns the symbol of the token, usually a shorter version of the
       * name.
       */
      function symbol() public view virtual override returns (string memory) {
          return _symbol;
      }
  
      /**
       * @dev Returns the number of decimals used to get its user representation.
       * For example, if `decimals` equals `2`, a balance of `505` tokens should
       * be displayed to a user as `5.05` (`505 / 10 ** 2`).
       *
       * Tokens usually opt for a value of 18, imitating the relationship between
       * Ether and Wei. This is the default value returned by this function, unless
       * it's overridden.
       *
       * NOTE: This information is only used for _display_ purposes: it in
       * no way affects any of the arithmetic of the contract, including
       * {IERC20-balanceOf} and {IERC20-transfer}.
       */
      function decimals() public view virtual override returns (uint8) {
          return 18;
      }
  
      /**
       * @dev See {IERC20-totalSupply}.
       */
      function totalSupply() public view virtual override returns (uint256) {
          return _totalSupply;
      }
  
      /**
       * @dev See {IERC20-balanceOf}.
       */
      function balanceOf(address account) public view virtual override returns (uint256) {
          return _balances[account];
      }
  
      /**
       * @dev See {IERC20-transfer}.
       *
       * Requirements:
       *
       * - `to` cannot be the zero address.
       * - the caller must have a balance of at least `amount`.
       */
      function transfer(address to, uint256 amount) public virtual override returns (bool) {
          address owner = _msgSender();
          _transfer(owner, to, amount);
          return true;
      }
  
      /**
       * @dev See {IERC20-allowance}.
       */
      function allowance(address owner, address spender) public view virtual override returns (uint256) {
          return _allowances[owner][spender];
      }
  
      /**
       * @dev See {IERC20-approve}.
       *
       * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
       * `transferFrom`. This is semantically equivalent to an infinite approval.
       *
       * Requirements:
       *
       * - `spender` cannot be the zero address.
       */
      function approve(address spender, uint256 amount) public virtual override returns (bool) {
          address owner = _msgSender();
          _approve(owner, spender, amount);
          return true;
      }
  
      /**
       * @dev See {IERC20-transferFrom}.
       *
       * Emits an {Approval} event indicating the updated allowance. This is not
       * required by the EIP. See the note at the beginning of {ERC20}.
       *
       * NOTE: Does not update the allowance if the current allowance
       * is the maximum `uint256`.
       *
       * Requirements:
       *
       * - `from` and `to` cannot be the zero address.
       * - `from` must have a balance of at least `amount`.
       * - the caller must have allowance for ``from``'s tokens of at least
       * `amount`.
       */
      function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
          address spender = _msgSender();
          _spendAllowance(from, spender, amount);
          _transfer(from, to, amount);
          return true;
      }
  
      /**
       * @dev Atomically increases the allowance granted to `spender` by the caller.
       *
       * This is an alternative to {approve} that can be used as a mitigation for
       * problems described in {IERC20-approve}.
       *
       * Emits an {Approval} event indicating the updated allowance.
       *
       * Requirements:
       *
       * - `spender` cannot be the zero address.
       */
      function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
          address owner = _msgSender();
          _approve(owner, spender, allowance(owner, spender) + addedValue);
          return true;
      }
  
      /**
       * @dev Atomically decreases the allowance granted to `spender` by the caller.
       *
       * This is an alternative to {approve} that can be used as a mitigation for
       * problems described in {IERC20-approve}.
       *
       * Emits an {Approval} event indicating the updated allowance.
       *
       * Requirements:
       *
       * - `spender` cannot be the zero address.
       * - `spender` must have allowance for the caller of at least
       * `subtractedValue`.
       */
      function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
          address owner = _msgSender();
          uint256 currentAllowance = allowance(owner, spender);
          require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
          unchecked {
              _approve(owner, spender, currentAllowance - subtractedValue);
          }
  
          return true;
      }
  
      /**
       * @dev Moves `amount` of tokens from `from` to `to`.
       *
       * This internal function is equivalent to {transfer}, and can be used to
       * e.g. implement automatic token fees, slashing mechanisms, etc.
       *
       * Emits a {Transfer} event.
       *
       * Requirements:
       *
       * - `from` cannot be the zero address.
       * - `to` cannot be the zero address.
       * - `from` must have a balance of at least `amount`.
       */
      function _transfer(address from, address to, uint256 amount) internal virtual {
          require(from != address(0), "ERC20: transfer from the zero address");
          require(to != address(0), "ERC20: transfer to the zero address");
  
          _beforeTokenTransfer(from, to, amount);
  
          uint256 fromBalance = _balances[from];
          require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
          unchecked {
              _balances[from] = fromBalance - amount;
              // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
              // decrementing then incrementing.
              _balances[to] += amount;
          }
  
          emit Transfer(from, to, amount);
  
          _afterTokenTransfer(from, to, amount);
      }
  
      /** @dev Creates `amount` tokens and assigns them to `account`, increasing
       * the total supply.
       *
       * Emits a {Transfer} event with `from` set to the zero address.
       *
       * Requirements:
       *
       * - `account` cannot be the zero address.
       */
      function _mint(address account, uint256 amount) internal virtual {
          require(account != address(0), "ERC20: mint to the zero address");
  
          _beforeTokenTransfer(address(0), account, amount);
  
          _totalSupply += amount;
          unchecked {
              // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
              _balances[account] += amount;
          }
          emit Transfer(address(0), account, amount);
  
          _afterTokenTransfer(address(0), account, amount);
      }
  
      /**
       * @dev Destroys `amount` tokens from `account`, reducing the
       * total supply.
       *
       * Emits a {Transfer} event with `to` set to the zero address.
       *
       * Requirements:
       *
       * - `account` cannot be the zero address.
       * - `account` must have at least `amount` tokens.
       */
      function _burn(address account, uint256 amount) internal virtual {
          require(account != address(0), "ERC20: burn from the zero address");
  
          _beforeTokenTransfer(account, address(0), amount);
  
          uint256 accountBalance = _balances[account];
          require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
          unchecked {
              _balances[account] = accountBalance - amount;
              // Overflow not possible: amount <= accountBalance <= totalSupply.
              _totalSupply -= amount;
          }
  
          emit Transfer(account, address(0), amount);
  
          _afterTokenTransfer(account, address(0), amount);
      }
  
      /**
       * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
       *
       * This internal function is equivalent to `approve`, and can be used to
       * e.g. set automatic allowances for certain subsystems, etc.
       *
       * Emits an {Approval} event.
       *
       * Requirements:
       *
       * - `owner` cannot be the zero address.
       * - `spender` cannot be the zero address.
       */
      function _approve(address owner, address spender, uint256 amount) internal virtual {
          require(owner != address(0), "ERC20: approve from the zero address");
          require(spender != address(0), "ERC20: approve to the zero address");
  
          _allowances[owner][spender] = amount;
          emit Approval(owner, spender, amount);
      }
  
      /**
       * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
       *
       * Does not update the allowance amount in case of infinite allowance.
       * Revert if not enough allowance is available.
       *
       * Might emit an {Approval} event.
       */
      function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
          uint256 currentAllowance = allowance(owner, spender);
          if (currentAllowance != type(uint256).max) {
              require(currentAllowance >= amount, "ERC20: insufficient allowance");
              unchecked {
                  _approve(owner, spender, currentAllowance - amount);
              }
          }
      }
  
      /**
       * @dev Hook that is called before any transfer of tokens. This includes
       * minting and burning.
       *
       * Calling conditions:
       *
       * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
       * will be transferred to `to`.
       * - when `from` is zero, `amount` tokens will be minted for `to`.
       * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
       * - `from` and `to` are never both zero.
       *
       * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
       */
      function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual {}
  
      /**
       * @dev Hook that is called after any transfer of tokens. This includes
       * minting and burning.
       *
       * Calling conditions:
       *
       * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
       * has been transferred to `to`.
       * - when `from` is zero, `amount` tokens have been minted for `to`.
       * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
       * - `from` and `to` are never both zero.
       *
       * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
       */
      function _afterTokenTransfer(address from, address to, uint256 amount) internal virtual {}
  }
  
  // lib/tokenized-strategy-periphery/lib/yearn-vaults-v3/contracts/interfaces/IVault.sol
  
  interface IVault is IERC4626 {
      // STRATEGY EVENTS
      event StrategyChanged(address indexed strategy, uint256 change_type);
      event StrategyReported(
          address indexed strategy,
          uint256 gain,
          uint256 loss,
          uint256 current_debt,
          uint256 protocol_fees,
          uint256 total_fees,
          uint256 total_refunds
      );
      // DEBT MANAGEMENT EVENTS
      event DebtUpdated(
          address indexed strategy,
          uint256 current_debt,
          uint256 new_debt
      );
      // ROLE UPDATES
      event RoleSet(address indexed account, uint256 role);
      event UpdateRoleManager(address indexed role_manager);
  
      event UpdateAccountant(address indexed accountant);
      event UpdateDefaultQueue(address[] new_default_queue);
      event UpdateUseDefaultQueue(bool use_default_queue);
      event UpdatedMaxDebtForStrategy(
          address indexed sender,
          address indexed strategy,
          uint256 new_debt
      );
      event UpdateDepositLimit(uint256 deposit_limit);
      event UpdateMinimumTotalIdle(uint256 minimum_total_idle);
      event UpdateProfitMaxUnlockTime(uint256 profit_max_unlock_time);
      event DebtPurchased(address indexed strategy, uint256 amount);
      event Shutdown();
  
      struct StrategyParams {
          uint256 activation;
          uint256 last_report;
          uint256 current_debt;
          uint256 max_debt;
      }
  
      function FACTORY() external view returns (uint256);
  
      function strategies(address) external view returns (StrategyParams memory);
  
      function default_queue(uint256) external view returns (address);
  
      function use_default_queue() external view returns (bool);
  
      function minimum_total_idle() external view returns (uint256);
  
      function deposit_limit() external view returns (uint256);
  
      function deposit_limit_module() external view returns (address);
  
      function withdraw_limit_module() external view returns (address);
  
      function accountant() external view returns (address);
  
      function roles(address) external view returns (uint256);
  
      function role_manager() external view returns (address);
  
      function future_role_manager() external view returns (address);
  
      function isShutdown() external view returns (bool);
  
      function nonces(address) external view returns (uint256);
  
      function initialize(
          address,
          string memory,
          string memory,
          address,
          uint256
      ) external;
  
      function set_accountant(address new_accountant) external;
  
      function set_default_queue(address[] memory new_default_queue) external;
  
      function set_use_default_queue(bool) external;
  
      function set_deposit_limit(uint256 deposit_limit) external;
  
      function set_deposit_limit(
          uint256 deposit_limit,
          bool should_override
      ) external;
  
      function set_deposit_limit_module(
          address new_deposit_limit_module
      ) external;
  
      function set_deposit_limit_module(
          address new_deposit_limit_module,
          bool should_override
      ) external;
  
      function set_withdraw_limit_module(
          address new_withdraw_limit_module
      ) external;
  
      function set_minimum_total_idle(uint256 minimum_total_idle) external;
  
      function setProfitMaxUnlockTime(
          uint256 new_profit_max_unlock_time
      ) external;
  
      function set_role(address account, uint256 role) external;
  
      function add_role(address account, uint256 role) external;
  
      function remove_role(address account, uint256 role) external;
  
      function transfer_role_manager(address role_manager) external;
  
      function accept_role_manager() external;
  
      function unlockedShares() external view returns (uint256);
  
      function pricePerShare() external view returns (uint256);
  
      function get_default_queue() external view returns (address[] memory);
  
      function process_report(
          address strategy
      ) external returns (uint256, uint256);
  
      function buy_debt(address strategy, uint256 amount) external;
  
      function add_strategy(address new_strategy) external;
  
      function revoke_strategy(address strategy) external;
  
      function force_revoke_strategy(address strategy) external;
  
      function update_max_debt_for_strategy(
          address strategy,
          uint256 new_max_debt
      ) external;
  
      function update_debt(
          address strategy,
          uint256 target_debt
      ) external returns (uint256);
  
      function update_debt(
          address strategy,
          uint256 target_debt,
          uint256 max_loss
      ) external returns (uint256);
  
      function shutdown_vault() external;
  
      function totalIdle() external view returns (uint256);
  
      function totalDebt() external view returns (uint256);
  
      function apiVersion() external view returns (string memory);
  
      function assess_share_of_unrealised_losses(
          address strategy,
          uint256 assets_needed
      ) external view returns (uint256);
  
      function profitMaxUnlockTime() external view returns (uint256);
  
      function fullProfitUnlockDate() external view returns (uint256);
  
      function profitUnlockingRate() external view returns (uint256);
  
      function lastProfitUpdate() external view returns (uint256);
  
      //// NON-STANDARD ERC-4626 FUNCTIONS \\\\
  
      function withdraw(
          uint256 assets,
          address receiver,
          address owner,
          uint256 max_loss
      ) external returns (uint256);
  
      function withdraw(
          uint256 assets,
          address receiver,
          address owner,
          uint256 max_loss,
          address[] memory strategies
      ) external returns (uint256);
  
      function redeem(
          uint256 shares,
          address receiver,
          address owner,
          uint256 max_loss
      ) external returns (uint256);
  
      function redeem(
          uint256 shares,
          address receiver,
          address owner,
          uint256 max_loss,
          address[] memory strategies
      ) external returns (uint256);
  
      function maxWithdraw(
          address owner,
          uint256 max_loss
      ) external view returns (uint256);
  
      function maxWithdraw(
          address owner,
          uint256 max_loss,
          address[] memory strategies
      ) external view returns (uint256);
  
      function maxRedeem(
          address owner,
          uint256 max_loss
      ) external view returns (uint256);
  
      function maxRedeem(
          address owner,
          uint256 max_loss,
          address[] memory strategies
      ) external view returns (uint256);
  
      //// NON-STANDARD ERC-20 FUNCTIONS \\\\
  
      function DOMAIN_SEPARATOR() external view returns (bytes32);
  
      function permit(
          address owner,
          address spender,
          uint256 amount,
          uint256 deadline,
          uint8 v,
          bytes32 r,
          bytes32 s
      ) external returns (bool);
  }
  
  // lib/tokenized-strategy/src/interfaces/ITokenizedStrategy.sol
  
  // Interface that implements the 4626 standard and the implementation functions
  interface ITokenizedStrategy is IERC4626, IERC20Permit {
      /*//////////////////////////////////////////////////////////////
                                   EVENTS
      //////////////////////////////////////////////////////////////*/
  
      event StrategyShutdown();
  
      event NewTokenizedStrategy(
          address indexed strategy,
          address indexed asset,
          string apiVersion
      );
  
      event Reported(
          uint256 profit,
          uint256 loss,
          uint256 protocolFees,
          uint256 performanceFees
      );
  
      event UpdatePerformanceFeeRecipient(
          address indexed newPerformanceFeeRecipient
      );
  
      event UpdateKeeper(address indexed newKeeper);
  
      event UpdatePerformanceFee(uint16 newPerformanceFee);
  
      event UpdateManagement(address indexed newManagement);
  
      event UpdateEmergencyAdmin(address indexed newEmergencyAdmin);
  
      event UpdateProfitMaxUnlockTime(uint256 newProfitMaxUnlockTime);
  
      event UpdatePendingManagement(address indexed newPendingManagement);
  
      /*//////////////////////////////////////////////////////////////
                             INITIALIZATION
      //////////////////////////////////////////////////////////////*/
  
      function initialize(
          address _asset,
          string memory _name,
          address _management,
          address _performanceFeeRecipient,
          address _keeper
      ) external;
  
      /*//////////////////////////////////////////////////////////////
                      NON-STANDARD 4626 OPTIONS
      //////////////////////////////////////////////////////////////*/
  
      function withdraw(
          uint256 assets,
          address receiver,
          address owner,
          uint256 maxLoss
      ) external returns (uint256);
  
      function redeem(
          uint256 shares,
          address receiver,
          address owner,
          uint256 maxLoss
      ) external returns (uint256);
  
      /*//////////////////////////////////////////////////////////////
                          MODIFIER HELPERS
      //////////////////////////////////////////////////////////////*/
  
      function requireManagement(address _sender) external view;
  
      function requireKeeperOrManagement(address _sender) external view;
  
      function requireEmergencyAuthorized(address _sender) external view;
  
      /*//////////////////////////////////////////////////////////////
                          KEEPERS FUNCTIONS
      //////////////////////////////////////////////////////////////*/
  
      function tend() external;
  
      function report() external returns (uint256 _profit, uint256 _loss);
  
      /*//////////////////////////////////////////////////////////////
                          CONSTANTS
      //////////////////////////////////////////////////////////////*/
  
      function MAX_FEE() external view returns (uint16);
  
      function FACTORY() external view returns (address);
  
      /*//////////////////////////////////////////////////////////////
                              GETTERS
      //////////////////////////////////////////////////////////////*/
  
      function apiVersion() external view returns (string memory);
  
      function pricePerShare() external view returns (uint256);
  
      function management() external view returns (address);
  
      function pendingManagement() external view returns (address);
  
      function keeper() external view returns (address);
  
      function emergencyAdmin() external view returns (address);
  
      function performanceFee() external view returns (uint16);
  
      function performanceFeeRecipient() external view returns (address);
  
      function fullProfitUnlockDate() external view returns (uint256);
  
      function profitUnlockingRate() external view returns (uint256);
  
      function profitMaxUnlockTime() external view returns (uint256);
  
      function lastReport() external view returns (uint256);
  
      function isShutdown() external view returns (bool);
  
      function unlockedShares() external view returns (uint256);
  
      /*//////////////////////////////////////////////////////////////
                              SETTERS
      //////////////////////////////////////////////////////////////*/
  
      function setPendingManagement(address) external;
  
      function acceptManagement() external;
  
      function setKeeper(address _keeper) external;
  
      function setEmergencyAdmin(address _emergencyAdmin) external;
  
      function setPerformanceFee(uint16 _performanceFee) external;
  
      function setPerformanceFeeRecipient(
          address _performanceFeeRecipient
      ) external;
  
      function setProfitMaxUnlockTime(uint256 _profitMaxUnlockTime) external;
  
      function shutdownStrategy() external;
  
      function emergencyWithdraw(uint256 _amount) external;
  }
  
  // lib/tokenized-strategy/src/interfaces/IStrategy.sol
  
  interface IStrategy is IBaseStrategy, ITokenizedStrategy {}
  
  // lib/tokenized-strategy-periphery/src/AprOracle/AprOracle.sol
  
  interface IOracle {
      function aprAfterDebtChange(
          address _strategy,
          int256 _delta
      ) external view returns (uint256);
  }
  
  /**
   *  @title APR Oracle
   *  @author Yearn.finance
   *  @dev Contract to easily retrieve the APR's of V3 vaults and
   *  strategies.
   *
   *  Can be used to check the current APR of any vault or strategy
   *  based on the current profit unlocking rate. As well as the
   *  expected APR given some change in totalAssets.
   *
   *  This can also be used to retrieve the expected APR a strategy
   *  is making, thats yet to be reported, if a strategy specific
   *  oracle has been added.
   *
   *  NOTE: All values are just at the specific time called and subject
   *  to change.
   */
  contract AprOracle is Governance {
      // Mapping of a strategy to its specific apr oracle.
      mapping(address => address) public oracles;
  
      // Used to get the Current and Expected APR'S.
      uint256 internal constant MAX_BPS_EXTENDED = 1_000_000_000_000;
      uint256 internal constant SECONDS_PER_YEAR = 31_556_952;
  
      constructor(address _governance) Governance(_governance) {}
  
      /**
       * @notice Get the current APR a strategy is earning.
       * @dev Will revert if an oracle has not been set for that strategy.
       *
       * This will be different than the {getExpectedApr()} which returns
       * the current APR based off of previously reported profits that
       * are currently unlocking.
       *
       * This will return the APR the strategy is currently earning that
       * has yet to be reported.
       *
       * @param _strategy Address of the strategy to check.
       * @param _debtChange Positive or negative change in debt.
       * @return apr The expected APR it will be earning represented as 1e18.
       */
      function getStrategyApr(
          address _strategy,
          int256 _debtChange
      ) public view virtual returns (uint256 apr) {
          // Get the oracle set for this specific strategy.
          address oracle = oracles[_strategy];
  
          // Don't revert if a oracle is not set.
          if (oracle != address(0)) {
              return IOracle(oracle).aprAfterDebtChange(_strategy, _debtChange);
          }
      }
  
      /**
       * @notice Get the current weighted APR of a strategy.
       * @dev Gives the apr weighted by its `totalAssets`. This can be used
       * to get the combined expected return of a collection of strategies.
       *
       * @param _strategy Address of the strategy.
       * @return . The current weighted APR of the strategy.
       */
      function weightedApr(
          address _strategy
      ) external view virtual returns (uint256) {
          return
              IStrategy(_strategy).totalAssets() * getStrategyApr(_strategy, 0);
      }
  
      /**
       * @notice Set a custom APR `_oracle` for a `_strategy`.
       * @dev Can only be called by the oracle's `governance` or
       *  management of the `_strategy`.
       *
       * The `_oracle` will need to implement the IOracle interface.
       *
       * @param _strategy Address of the strategy.
       * @param _oracle Address of the APR Oracle.
       */
      function setOracle(address _strategy, address _oracle) external virtual {
          if (governance != msg.sender) {
              require(
                  msg.sender == IStrategy(_strategy).management(),
                  "!authorized"
              );
          }
  
          oracles[_strategy] = _oracle;
      }
  
      /**
       * @notice Get the current APR for a V3 vault or strategy.
       * @dev This returns the current APR based off the current
       * rate of profit unlocking for either a vault or strategy.
       *
       * Will return 0 if there is no profit unlocking or no assets.
       *
       * @param _vault The address of the vault or strategy.
       * @return apr The current apr expressed as 1e18.
       */
      function getCurrentApr(
          address _vault
      ) external view virtual returns (uint256 apr) {
          return getExpectedApr(_vault, 0);
      }
  
      /**
       * @notice Get the expected APR for a V3 vault or strategy based on `_delta`.
       * @dev This returns the expected APR based off the current
       * rate of profit unlocking for either a vault or strategy
       * given some change in the total assets.
       *
       * Will return 0 if there is no profit unlocking or no assets.
       *
       * This can be used to predict the change in current apr given some
       * deposit or withdraw to the vault.
       *
       * @param _vault The address of the vault or strategy.
       * @param _delta The positive or negative change in `totalAssets`.
       * @return apr The expected apr expressed as 1e18.
       */
      function getExpectedApr(
          address _vault,
          int256 _delta
      ) public view virtual returns (uint256 apr) {
          IVault vault = IVault(_vault);
  
          // Check if the full profit has already been unlocked.
          if (vault.fullProfitUnlockDate() <= block.timestamp) return 0;
  
          // Need the total assets in the vault post delta.
          uint256 assets = uint256(int256(vault.totalAssets()) + _delta);
  
          // No apr if there are no assets.
          if (assets == 0) return 0;
  
          // We need to get the amount of assets that are unlocking per second.
          // `profitUnlockingRate` is in shares so we convert it to assets.
          uint256 assetUnlockingRate = vault.convertToAssets(
              vault.profitUnlockingRate()
          );
  
          // APR = assets unlocking per second * seconds per year / the total assets.
          apr =
              (1e18 * assetUnlockingRate * SECONDS_PER_YEAR) /
              MAX_BPS_EXTENDED /
              assets;
      }
  }
  
  // src/periphery/StrategyAprOracle.sol
  
  contract StrategyAprOracle is AprOracleBase {
  
      AprOracle internal constant aprOracle = AprOracle(0x27aD2fFc74F74Ed27e1C0A19F1858dD0963277aE);
  
      mapping(address => address) public remapping;
  
      constructor() AprOracleBase("Generic strategy Oracle", msg.sender) {}
  
      /**
       * @notice Will return the expected Apr of a strategy post a debt change.
       * @dev _delta is a signed integer so that it can also represent a debt
       * decrease.
       *
       * This should return the annual expected return at the current timestamp
       * represented as 1e18.
       *
       *      ie. 10% == 1e17
       *
       * _delta will be == 0 to get the current apr.
       *
       * This will potentially be called during non-view functions so gas
       * efficiency should be taken into account.
       *
       * @param _strategy The token to get the apr for.
       * @param _delta The difference in debt.
       * @return . The expected apr for the strategy represented as 1e18.
       */
      function aprAfterDebtChange(
          address _strategy,
          int256 _delta
      ) external view override returns (uint256) {
          address _remapping = remapping[_strategy];
          if (_remapping != address(0)) _strategy = _remapping;
  
          return aprOracle.getExpectedApr(_strategy, _delta);
      }
  
      function setRemapping(address _og, address _ng) external onlyGovernance {
          remapping[_og] = _ng;
      }
  }